

Finite Automata using the Hereditarily Finite Sets

Prof. Lawrence C Paulson
Computer Laboratory, University of Cambridge

Abstract

Finite Automata, both deterministic and non-deterministic, for regular languages. The Myhill-Nerode Theorem. Closure under intersection, concatenation, etc. Regular expressions define regular languages. Closure under reversal; the powerset construction mapping NFAs to DFAs. Left and right languages; minimal DFAs. Brzozowski's minimization algorithm. Uniqueness up to isomorphism of minimal DFAs.

Chapter 1

Finite Automata using the Hereditarily Finite Sets

```
theory Finite_Automata_HF imports
  HereditarilyFinite.Ordinal
  "Regular-Sets.Regular_Exp"
begin
```

Finite Automata, both deterministic and non-deterministic, for regular languages. The Myhill-Nerode Theorem. Closure under intersection, concatenation, etc. Regular expressions define regular languages. Closure under reversal; the powerset construction mapping NFAs to DFAs. Left and right languages; minimal DFAs. Brzozowski's minimization algorithm. Uniqueness up to isomorphism of minimal DFAs.

Initially this formalization was based on automata whose state type was always `hf`. In a revision, it was generalized: many of the constructions are now based on automata with a polymorphic state type `'s`.

1.1 Deterministic Finite Automata

Right invariance is the key property for equivalence relations on states of DFAs.

```
definition right_invariant :: "('a list × 'a list) set ⇒ bool" where
  "right_invariant r ≡ (∀ u v w. (u,v) ∈ r ⟶ (u@w, v@w) ∈ r)"
```

1.1.1 Basic Definitions

We try to make as much as possible polymorphic in the state space `'s` and independent of type `hf` to increase reusability.

First, the record for DFAs

```

record ('a,'s) dfa = states :: "'s set"
                    init   :: "'s"
                    final  :: "'s set"
                    nxt    :: "'s  $\Rightarrow$  'a  $\Rightarrow$  's"

locale dfa =
  fixes M :: "('a,'s) dfa"
  assumes init [simp]: "init M  $\in$  states M"
           and final:   "final M  $\subseteq$  states M"
           and nxt:     " $\bigwedge$ q x. q  $\in$  states M  $\implies$  nxt M q x  $\in$  states M"
           and finite:  "finite (states M)"
begin

lemma finite_final [simp]: "finite (final M)"
  <proof>

Transition function for a given starting state and word.
primrec nextl :: "[ 's, 'a list ]  $\Rightarrow$  's" where
  "nextl q []      = q"
  | "nextl q (x#xs) = nextl (nxt M q x) xs"

definition language :: "'a list set" where
  "language  $\equiv$  {xs. nextl (init M) xs  $\in$  final M}"

The left language WRT a state q is the set of words that lead to q.
definition left_lang :: "'s  $\Rightarrow$  'a list set" where
  "left_lang q  $\equiv$  {u. nextl (init M) u = q}"

Part of Prop 1 of Jean-Marc Champarnaud, A. Khorsi and T. Paranthoën,
Split and join for minimizing: Brzozowski's algorithm, Prague Stringology
Conference 2002

lemma left_lang_disjoint:
  "q1  $\neq$  q2  $\implies$  left_lang q1  $\cap$  left_lang q2 = {}"
  <proof>

The right language WRT a state q is the set of words that go from q to F.
definition right_lang :: "'s  $\Rightarrow$  'a list set" where
  "right_lang q  $\equiv$  {u. nextl q u  $\in$  final M}"

lemma language_eq_right_lang: "language = right_lang (init M)"
  <proof>

lemma nextl_app: "nextl q (xs@ys) = nextl (nextl q xs) ys"
  <proof>

lemma nextl_snoc [simp]: "nextl q (xs@[x]) = nxt M (nextl q xs) x"
  <proof>

```

lemma nextl_state: "q ∈ states M ⇒ nextl q xs ∈ states M"
⟨proof⟩

lemma nextl_init_state [simp]: "nextl (init M) xs ∈ states M"
⟨proof⟩

1.1.2 An Equivalence Relation on States

Two words are equivalent if they take the machine to the same state. See e.g. Kozen, Automata and Computability, Springer, 1997, page 90.

This relation asks, do u and v lead to the same state?

definition eq_nextl :: "('a list × 'a list) set" where
"eq_nextl ≡ {(u,v). nextl (init M) u = nextl (init M) v}"

lemma equiv_eq_nextl: "equiv UNIV eq_nextl"
⟨proof⟩

lemma right_invariant_eq_nextl: "right_invariant eq_nextl"
⟨proof⟩

lemma range_nextl: "range (nextl (init M)) ⊆ states M"
⟨proof⟩

lemma eq_nextl_class_in_left_lang_im: "eq_nextl ‘‘ {u} ∈ left_lang ‘
states M"
⟨proof⟩

lemma language_eq_nextl: "language = eq_nextl ‘‘ (⋃q ∈ final M. left_lang
q)"
⟨proof⟩

lemma finite_index_eq_nextl: "finite (UNIV // eq_nextl)"
⟨proof⟩

lemma index_eq_nextl_le_states: "card (UNIV // eq_nextl) ≤ card (states
M)"
⟨proof⟩

1.1.3 Minimisation via Accessibility

definition accessible :: "'s set" where
"accessible ≡ {q. left_lang q ≠ {}}"

lemma accessible_imp_states: "q ∈ accessible ⇒ q ∈ states M"
⟨proof⟩

lemma nxt_accessible: "q ∈ accessible ⇒ nxt M q a ∈ accessible"
⟨proof⟩

```

lemma inj_on_left_lang: "inj_on left_lang accessible"
  <proof>

definition path_to :: "'s ⇒ 'a list" where
  "path_to q ≡ SOME u. u ∈ left_lang q"

lemma path_to_left_lang: "q ∈ accessible ⇒ path_to q ∈ left_lang q"
  <proof>

lemma nextl_path_to: "q ∈ accessible ⇒ nextl (dfa.init M) (path_to
q) = q"
  <proof>

definition Accessible_dfa :: "('a,'s) dfa" where
  "Accessible_dfa = (dfa.states = accessible,
    init = init M,
    final = final M ∩ accessible,
    nxt = nxt M)"

lemma states_Accessible_dfa [simp]: "states Accessible_dfa = accessible"
  <proof>

lemma init_Accessible_dfa [simp]: "init Accessible_dfa = init M"
  <proof>

lemma final_Accessible_dfa [simp]: "final Accessible_dfa = final M ∩
accessible"
  <proof>

lemma nxt_Accessible_dfa [simp]: "nxt Accessible_dfa = nxt M"
  <proof>

interpretation Accessible: dfa Accessible_dfa
  <proof>

lemma dfa_Accessible: "dfa Accessible_dfa"
  <proof>

lemma nextl_Accessible_dfa [simp]:
  "q ∈ accessible ⇒ Accessible.nextl q u = nextl q u"
  <proof>

lemma init_Accessible: "init M ∈ accessible"
  <proof>

declare nextl_Accessible_dfa [OF init_Accessible, simp]

lemma Accessible_left_lang_eq [simp]: "Accessible.left_lang q = left_lang

```

```

q"
  ⟨proof⟩

lemma Accessible_right_lang_eq [simp]:
  "q ∈ accessible ⇒ Accessible.right_lang q = right_lang q"
  ⟨proof⟩

lemma Accessible_language [simp]: "Accessible.language = language"
  ⟨proof⟩

lemma Accessible_accessible [simp]: "Accessible.accessible = accessible"
  ⟨proof⟩

lemma left_lang_half:
  assumes sb: " $\bigcup (\text{left\_lang } ' qs1) \subseteq \bigcup (\text{left\_lang } ' qs2)$ "
    and ne: " $\bigwedge x. x \in qs1 \Rightarrow \text{left\_lang } x \neq \{\}$ "
    shows " $qs1 \subseteq qs2$ "
  ⟨proof⟩

lemma left_lang_UN:
  " $[\bigcup (\text{left\_lang } ' qs1) = \bigcup (\text{left\_lang } ' qs2); qs1 \cup qs2 \subseteq \text{accessible}]$ 
 $\Rightarrow qs1 = qs2$ "
  ⟨proof⟩

definition minimal where
  "minimal  $\equiv$  accessible = states M  $\wedge$  inj_on right_lang (dfa.states M)"

1.1.4 An Equivalence Relation on States

Collapsing map on states. Two states are equivalent if they yield identical
outcomes

definition eq_right_lang :: "('s × 's) set" where
  "eq_right_lang  $\equiv$  {(u,v). u ∈ states M  $\wedge$  v ∈ states M  $\wedge$  right_lang u
= right_lang v}"

lemma equiv_eq_right_lang: "equiv (states M) eq_right_lang"
  ⟨proof⟩

lemma eq_right_lang_finite_index: "finite (states M // eq_right_lang)"
  ⟨proof⟩

end

Now we need to specialize to hf states.

type_synonym 'a dfa_hf = "('a,hf) dfa"

locale dfa_hf = dfa M for M :: "'a dfa_hf"

```

```

begin

definition Collapse_dfa :: "'a dfa_hf" where
  "Collapse_dfa = (dfa.states = HF ' (states M // eq_right_lang),
    init      = HF (eq_right_lang ' ' {init M}),
    final     = {HF (eq_right_lang ' ' {q}) | q. q ∈ final
M},
    nxt       = λQ x. HF (⋃q ∈ hfset Q. eq_right_lang
' ' {nxt M q x})))"

lemma nxt_Collapse_resp: "(λq. eq_right_lang ' ' {nxt M q x}) respects
eq_right_lang"
  <proof>

lemma finite_Collapse_state: "Q ∈ states M // eq_right_lang ⇒ finite
Q"
  <proof>

interpretation Collapse: dfa Collapse_dfa
  <proof>

corollary dfa_Collapse: "dfa Collapse_dfa"
  <proof>

lemma nextl_Collapse_dfa:
  "Q = HF (eq_right_lang ' ' {q}) ⇒ Q ∈ dfa.states Collapse_dfa ⇒
  q ∈ states M ⇒
  Collapse.nextl Q u = HF (eq_right_lang ' ' {nextl q u})"
  <proof>

lemma ext_language_Collapse_dfa:
  "u ∈ Collapse.language ↔ u ∈ language"
  <proof>

theorem language_Collapse_dfa:
  "Collapse.language = language"
  <proof>

lemma card_Collapse_dfa: "card (states M // eq_right_lang) ≤ card (states
M)"
  <proof>

end

```

1.1.5 Isomorphisms Between DFAs

```

locale dfa_isomorphism = M: dfa M + N: dfa N for M :: "('a,'sm) dfa" and
N :: "('a,'sn) dfa" +
  fixes h :: "'sm ⇒ 'sn"

```

```

    assumes h: "bij_betw h (states M) (states N)"
    and init [simp]: "h (init M) = init N"
    and final [simp]: "h ' final M = final N"
    and nxt [simp]: " $\bigwedge q x. q \in \text{states } M \implies h (\text{nxt } M \ q \ x) = \text{nxt } N \ (h \ q) \ x$ "

```

begin

```

lemma nextl [simp]: "q ∈ states M ⇒ h (M.nextl q u) = N.nextl (h q) u"
  <proof>

```

```

theorem language: "M.language = N.language"
  <proof>

```

```

lemma nxt_inv_into: "q ∈ states N ⇒ nxt N q x = h (nxt M (inv_into (states M) h q) x)"
  <proof>

```

```

lemma sym: "dfa_isomorphism N M (inv_into (states M) h)"
  <proof>

```

```

lemma trans: "dfa_isomorphism N N' h' ⇒ dfa_isomorphism M N' (h' o h)"
  <proof>

```

end

In order to transition between 's and hf, we use bijections:

```

lemma inj_on_finite_hf:
  <finite S ⇒ ∃f:: 's ⇒ hf. inj_on f S>
  <proof>

```

```

lemma bij_betw_finite_hf:
  <finite S ⇒ ∃f:: 's ⇒ hf. bij_betw f S (f ' S)>
  <proof>

```

There is always an isomorphism between a ('a, 's) dfa and a 'a dfa_hf:

```

context dfa
begin

```

```

definition bij_s_hf :: "'s ⇒ hf" where
  "bij_s_hf = (SOME f :: 's ⇒ hf. bij_betw f (dfa.states M) (f ' dfa.states M))"

```

```

lemma bij_betw_bij_s_hf: "bij_betw bij_s_hf (dfa.states M) (bij_s_hf ' dfa.states M)"
  <proof>

```

```

abbreviation bij_s_hf_M :: "'a dfa_hf" where
  "bij_s_hf_M  $\equiv$  ( $\lfloor$  dfa.states = bij_s_hf ' dfa.states M,
    dfa.init = bij_s_hf (dfa.init M),
    dfa.final = bij_s_hf ' dfa.final M,
    dfa.nxt = ( $\lambda$ q x. bij_s_hf (dfa.nxt M (the_inv_into (dfa.states
M) bij_s_hf q) x))  $\rfloor$ "

```

```

lemma dfa_bij_s_hf_M: "dfa bij_s_hf_M"
  <proof>

```

```

interpretation M_iso: dfa_isomorphism M bij_s_hf_M bij_s_hf
  <proof>

```

```

lemmas L_M_eq_L_bij_s_hf_M = M_iso.language

```

```

end

```

1.2 The Myhill-Nerode theorem: three characterisations of a regular language

```

definition regular :: "'a list set  $\Rightarrow$  bool" where
  "regular L  $\equiv$   $\exists$ M :: 'a dfa_hf. dfa M  $\wedge$  dfa.language M = L"

```

```

definition MyhillNerode :: "'a list set  $\Rightarrow$  ('a list * 'a list) set  $\Rightarrow$  bool"
where
  "MyhillNerode L R  $\equiv$  equiv UNIV R  $\wedge$  right_invariant R  $\wedge$  finite (UNIV//R)
 $\wedge$  ( $\exists$ A. L = R 'A)"

```

This relation can be seen as an abstraction of the idea, do u and v lead to the same state? Compare with eq_next1, which does precisely that.

```

definition eq_app_right :: "'a list set  $\Rightarrow$  ('a list * 'a list) set" where
  "eq_app_right L  $\equiv$  {(u,v).  $\forall$ w. u@w  $\in$  L  $\longleftrightarrow$  v@w  $\in$  L}"

```

```

lemma equiv_eq_app_right: "equiv UNIV (eq_app_right L)"
  <proof>

```

```

lemma right_invariant_eq_app_right: "right_invariant (eq_app_right L)"
  <proof>

```

```

lemma eq_app_right_eq: "eq_app_right L ' ' L = L"
  <proof>

```

```

lemma MN_eq_app_right:
  "finite (UNIV // eq_app_right L)  $\implies$  MyhillNerode L (eq_app_right
L)"
  <proof>

```

```

lemma MN_refines: "[MyhillNerode L R; (x,y)  $\in$  R]  $\implies$  x  $\in$  L  $\longleftrightarrow$  y  $\in$  L"

```

<proof>

lemma MN_refines_eq_app_right: "MyhillNerode L R \implies R \subseteq eq_app_right L"

<proof>

Step 1 in the circle of implications: every regular language L is recognised by some Myhill-Nerode relation, R

context dfa
begin

lemma regular_dfa: "regular language"

<proof>

lemma MN_eq_next1: "MyhillNerode language eq_next1"

<proof>

corollary eq_next1_refines_eq_app_right: "eq_next1 \subseteq eq_app_right language"

<proof>

lemma index_le_index_eq_next1:

"card (UNIV // eq_app_right language) \leq card (UNIV // eq_next1)"

<proof>

A specific lower bound on the number of states in a DFA

lemma index_eq_app_right_lower:

"card (UNIV // eq_app_right language) \leq card (states M)"

<proof>

end

lemma L1_2: "regular L \implies \exists R. MyhillNerode L R"

<proof>

Step 2: every Myhill-Nerode relation R for the language L can be mapped to the canonical M-N relation.

lemma L2_3:

assumes "MyhillNerode L R"

obtains "finite (UNIV // eq_app_right L)"

"card (UNIV // eq_app_right L) \leq card (UNIV // R)"

<proof>

Working towards step 3. Also, every Myhill-Nerode relation R for L can be mapped to a machine. The locale below constructs such a DFA.

locale MyhillNerode_dfa =

fixes L :: "'a list set" and R :: "('a list * 'a list) set"

and A :: "'a list set" and n :: nat and h :: "'a list set \Rightarrow hf"

assumes eqR: "equiv UNIV R"

and riR: "right_invariant R"

```

    and L: "L = R 'A"
    and h: "bij_betw h (UNIV//R) (hfset (ord_of n))"
begin

lemma injh: "inj_on h (UNIV//R)"
  <proof>

definition hinv (<math>h^{-1}</math>) where "h-1  $\equiv$  inv_into (UNIV//R) h"

lemma finix: "finite (UNIV//R)"
  <proof>

definition DFA :: "'a dfa_hf" where
  "DFA = (states = h ' (UNIV//R),
    init = h (R ' ' {[]}),
    final = {h (R ' ' {u}) | u. u  $\in$  A},
    nxt =  $\lambda$ q x. h ( $\bigcup$ u  $\in$  h-1 q. R ' ' {u@[x]}))"

lemma resp: " $\bigwedge$ x. ( $\lambda$ u. R ' ' {u @ [x]}) respects R"
  <proof>

lemma dfa: "dfa DFA"
  <proof>

interpretation MN: dfa DFA
  <proof>

lemma MyhillNerode: "MyhillNerode L R"
  <proof>

lemma R_iff: "( $\exists$ x $\in$ L. (u, x)  $\in$  R) = (u  $\in$  L)"
  <proof>

lemma next1: "MN.next1 (init DFA) u = h (R ' ' {u})"
  <proof>

lemma language: "MN.language = L"
  <proof>

lemma card_states: "card (states DFA) = card (UNIV // R)"
  <proof>

end

theorem MN_imp_dfa:
  assumes "MyhillNerode L R"
  obtains M where "dfa_hf M" "dfa.language M = L" "card (states M) =
card (UNIV//R)"

```

<proof>

corollary MN_imp_regular:

assumes "MyhillNerode L R" shows "regular L"

<proof>

corollary eq_app_right_finite_index_imp_dfa:

assumes "finite (UNIV // eq_app_right L)"

obtains M where

"dfa_hf M" "dfa.language M = L" "card (states M) = card (UNIV // eq_app_right L)"

<proof>

Step 3

corollary L3_1: "finite (UNIV // eq_app_right L) \implies regular L"

<proof>

1.3 Non-Deterministic Finite Automata

These NFAs may include epsilon-transitions and multiple start states.

1.3.1 Basic Definitions

```
record ('a,'s) nfa = states :: "'s set"
                  init   :: "'s set"
                  final  :: "'s set"
                  nxt    :: "'s  $\Rightarrow$  'a  $\Rightarrow$  's set"
                  eps    :: "('s * 's) set"
```

locale nfa =

fixes M :: "('a,'s) nfa"

assumes init: "init M \subseteq states M"

and final: "final M \subseteq states M"

and nxt: " $\bigwedge q x. q \in \text{states } M \implies \text{nxt } M \ q \ x \subseteq \text{states } M$ "

and finite: "finite (states M)"

begin

lemma subset_states_finite [intro,simp]: "Q \subseteq states M \implies finite Q"

<proof>

definition epsclo :: "'s set \Rightarrow 's set" where

"eps clo Q \equiv states M \cap ($\bigcup_{q \in Q. \{q'\}. (q,q') \in (\text{eps } M)^*$)"

lemma epsclo_eq_Image: "eps clo Q = states M \cap (eps M)* " Q"

<proof>

lemma epsclo_empty [simp]: "eps clo {} = {}"

<proof>

lemma epsclo_idem [simp]: "eps clo (eps clo Q) = eps clo Q"
 <proof>

lemma epsclo_increasing: "Q \cap states M \subseteq eps clo Q"
 <proof>

lemma epsclo_Un [simp]: "eps clo (Q1 \cup Q2) = eps clo Q1 \cup eps clo Q2"
 <proof>

lemma epsclo_UN [simp]: "eps clo ($\bigcup_{x \in A}. B x$) = ($\bigcup_{x \in A}. eps clo (B x)$)"
 <proof>

lemma epsclo_subset [simp]: "eps clo Q \subseteq states M"
 <proof>

lemma epsclo_trivial [simp]: "eps M \subseteq Q \times Q \implies eps clo Q = states M \cap Q"
 <proof>

lemma epsclo_mono: "Q' \subseteq Q \implies eps clo Q' \subseteq eps clo Q"
 <proof>

lemma finite_epsclo [simp]: "finite (eps clo Q)"
 <proof>

lemma finite_final: "finite (final M)"
 <proof>

lemma finite_nxt: "q \in states M \implies finite (nxt M q x)"
 <proof>

Transition function for a given starting state and word.

primrec nextl :: "'s set, 'a list] \Rightarrow 's set" where
 "nextl Q [] = eps clo Q"
 | "nextl Q (x#xs) = nextl ($\bigcup_{q \in eps clo Q}. nxt M q x$) xs"

definition language :: "'a list set" where
 "language \equiv {xs. nextl (init M) xs \cap final M \neq {}}"

The right language WRT a state q is the set of words that go from q to F.

definition right_lang :: "'s \Rightarrow 'a list set" where
 "right_lang q \equiv {u. nextl {q} u \cap final M \neq {}}"

lemma nextl_epsclo [simp]: "nextl (eps clo Q) xs = nextl Q xs"
 <proof>

lemma epsclo_nextl [simp]: "eps clo (nextl Q xs) = nextl Q xs"
 <proof>

```

lemma nextl_app: "nextl Q (xs@ys) = nextl (nextl Q xs) ys"
  <proof>

lemma nextl_snoc [simp]: "nextl Q (xs@[x]) = (⋃ q ∈ nextl Q xs. epsclo
(nxt M q x))"
  <proof>

lemma nextl_state: "nextl Q xs ⊆ states M"
  <proof>

lemma nextl_mono: "Q' ⊆ Q ⇒ nextl Q' u ⊆ nextl Q u"
  <proof>

lemma nextl_eps: "q ∈ nextl Q u ⇒ (q,q') ∈ eps M ⇒ q' ∈ states M
⇒ q' ∈ nextl Q u"
  <proof>

lemma finite_nextl: "finite (nextl Q u)"
  <proof>

lemma nextl_empty [simp]: "nextl {} xs = {}"
  <proof>

lemma nextl_Un: "nextl (Q1 ∪ Q2) xs = nextl Q1 xs ∪ nextl Q2 xs"
  <proof>

lemma nextl_UN: "nextl (⋃ i ∈ I. f i) xs = (⋃ i ∈ I. nextl (f i) xs)"
  <proof>

end

```

Also works for state type 's of DFA leading to state type 's set in NFA

```

lemma nfa_of_dfa:
  assumes "dfa (M::('a,hf)dfa)"
  obtains N :: "('a,hf)nfa" where "nfa N ∧ nfa.language N = dfa.language
M"
  <proof>

```

```

corollary nfa_if_regular:
  assumes "regular L"
  obtains N :: "('a,hf)nfa" where "nfa N ∧ nfa.language N = L"
  <proof>

```

1.3.2 The Powerset Construction

First: The construction of a ('a, 's set) dfa from an ('a, 's) nfa. Is not used later on but provides an easy means of showing regularity of some language by constructing an NFA without having to use hf.

```

context nfa
begin

definition Power_dfa :: "('a, 's set) dfa" where
  "Power_dfa = (dfa.states = {eps clo q | q. q ∈ Pow (states M)},
    init = eps clo (init M),
    final = {eps clo Q | Q. Q ⊆ states M ∧ Q ∩ final
M ≠ {}},
    nxt = λQ x. ∪ q ∈ eps clo Q. eps clo (nxt M q x))"

lemma states_Power_dfa [simp]: "dfa.states Power_dfa = eps clo ' Pow (states
M)"
  <proof>

lemma init_Power_dfa [simp]: "dfa.init Power_dfa = eps clo (nfa.init M)"
  <proof>

lemma final_Power_dfa [simp]: "dfa.final Power_dfa = {eps clo Q | Q. Q
⊆ states M ∧ Q ∩ final M ≠ {}}"
  <proof>

lemma nxt_Power_dfa [simp]: "dfa.nxt Power_dfa = (λQ x. ∪ q ∈ eps clo
Q. eps clo (nxt M q x))"
  <proof>

interpretation Power: dfa Power_dfa
  <proof>

The Power DFA accepts the same language as the NFA.

theorem Power_language [simp]: "Power.language = language"
  <proof>

Every language accepted by a NFA is also accepted by a DFA.

corollary imp_regular: "regular language"
  <proof>

end

```

As above, outside the locale

```

corollary nfa_imp_regular:
  assumes "nfa M" "nfa.language M = L"
  shows "regular L"
  <proof>

```

```

type_synonym 'a nfa_hf = "('a, hf) nfa"

```

The construction of a 'a dfa_hf from an 'a nfa_hf. Very little can be reused from the generic 's-based construction above.

locale nfa_hf = nfa M for M :: "'a nfa_hf"
begin

definition Power_dfa_hf :: "'a dfa_hf" where
 "Power_dfa_hf = (dfa.states = HF ' dfa.states Power_dfa,
 init = HF (dfa.init Power_dfa),
 final = HF ' dfa.final Power_dfa,
 nxt = $\lambda Q. HF \circ dfa.nxt \text{ Power_dfa } (hfset \ Q)$)"

lemma states_Power_dfa [simp]: "dfa.states Power_dfa_hf = HF ' epsclo
 ' Pow (states M)"
 <proof>

lemma init_Power_dfa [simp]: "dfa.init Power_dfa_hf = HF (dfa.init Power_dfa)"
 <proof>

lemma final_Power_dfa [simp]: "dfa.final Power_dfa_hf = {HF (eps clo Q)
 | Q. Q \subseteq states M \wedge Q \cap final M \neq {}}"
 <proof>

lemma nxt_Power_dfa [simp]: "dfa.nxt Power_dfa_hf = ($\lambda Q \ x. HF(\bigcup q \in$
 epsclo (hfset Q). eps clo (nxt M q x)))"
 <proof>

interpretation Power: dfa Power_dfa_hf
 <proof>

corollary dfa_Power: "dfa Power_dfa_hf"
 <proof>

lemma nextl_Power_dfa:
 "qs \in dfa.states Power_dfa_hf
 \implies dfa.nextl Power_dfa_hf qs u = HF ($\bigcup q \in hfset \ qs. nextl \ \{q\} \ u$)"
 <proof>

Part of Prop 4 of Jean-Marc Champarnaud, A. Khorsi and T. Paranthoën
 (2002)

lemma Power_right_lang:
 assumes "qs \in dfa.states Power_dfa_hf"
 shows "Power.right_lang qs = ($\bigcup q \in hfset \ qs. right_lang \ q$)"
 <proof>

The Power DFA accepts the same language as the NFA.

theorem Power_language [simp]: "Power.language = language"
 <proof>

end

1.4 Closure Properties for Regular Languages

1.4.1 The Empty Language

theorem regular_empty: "regular {}"
<proof>

1.4.2 The Empty Word

theorem regular_nullstr: "regular {[]}"
<proof>

1.4.3 Single Symbol Languages

theorem regular_singstr: "regular {[a]}"
<proof>

1.4.4 The Complement of a Language

theorem regular_Compl:
 assumes S: "regular S" **shows** "regular (-S)"
<proof>

1.4.5 The Intersection and Union of Two Languages

By the familiar product construction

theorem regular_Int:
 assumes S: "regular S" **and** T: "regular T" **shows** "regular (S \cap T)"
<proof>

corollary regular_Un:
 assumes S: "regular S" **and** T: "regular T" **shows** "regular (S \cup T)"
<proof>

1.4.6 The Concatenation of Two Languages

lemma Inlr_rtrancl [simp]: " $((\lambda q. (HF.Inl q, HF.Inr a)) ' A)^* = ((\lambda q. (HF.Inl q, HF.Inr a)) ' A)^*$ "
<proof>

theorem regular_conc:
 assumes S: "regular S" **and** T: "regular T" **shows** "regular (S @@ T)"
<proof>

lemma regular_word: "regular {u}"
<proof>

All finite sets are regular.

theorem regular_finite: "finite L \implies regular L"
<proof>

1.4.7 The Kleene Star of a Language

theorem regular_star:
 assumes S: "regular S" shows "regular (star S)"
 <proof>

1.4.8 The Reversal of a Regular Language

definition Reverse_nfa :: "'a dfa_hf \Rightarrow 'a nfa_hf" where
 "Reverse_nfa MS = (nfa.states = dfa.states MS,
 init = dfa.final MS,
 final = {dfa.init MS},
 nxt = $\lambda q x. \{q' \in \text{dfa.states MS}. q = \text{dfa.nxt MS } q' \ x\},$
 eps = {})"

lemma states_Reverse_nfa [simp]: "states (Reverse_nfa MS) = dfa.states MS"
 <proof>

lemma init_Reverse_nfa [simp]: "init (Reverse_nfa MS) = dfa.final MS"
 <proof>

lemma final_Reverse_nfa [simp]: "final (Reverse_nfa MS) = {dfa.init MS}"
 <proof>

lemma nxt_Reverse_nfa [simp]:
 "nxt (Reverse_nfa MS) q x = {q' \in dfa.states MS. q = dfa.nxt MS q' x}"
 <proof>

lemma eps_Reverse_nfa [simp]: "eps (Reverse_nfa MS) = {}"
 <proof>

context dfa_hf
begin

lemma nfa_Reverse_nfa: "nfa (Reverse_nfa M)"
 <proof>

lemma nextl_Reverse_nfa:
 "nfa.nextl (Reverse_nfa M) Q u = {q' \in dfa.states M. dfa.nextl M q' (rev u) \in Q}"
 <proof>

Part of Prop 3 of Jean-Marc Champarnaud, A. Khorsi and T. Paranthoën (2002)

lemma right_lang_Reverse: "nfa.right_lang (Reverse_nfa M) q = rev (dfa.left_lang M q)"
 <proof>

```

lemma right_lang_Reverse_disjoint:
  "q1 ≠ q2 ⇒ nfa.right_lang (Reverse_nfa M) q1 ∩ nfa.right_lang (Reverse_nfa
M) q2 = {}"
  ⟨proof⟩

lemma epsclo_Reverse_nfa [simp]: "nfa.esclo (Reverse_nfa M) Q = Q
∩ dfa.states M"
  ⟨proof⟩

theorem language_Reverse_nfa [simp]:
  "nfa.language (Reverse_nfa M) = (rev ‘ dfa.language M)"
  ⟨proof⟩

end

corollary regular_Reverse:
  assumes S: "regular S" shows "regular (rev ‘ S)"
  ⟨proof⟩

All regular expressions yield regular languages.

theorem regular_lang: "regular (lang r)"
  ⟨proof⟩

```

1.5 Brzowski’s Minimization Algorithm

```

context dfa_hf
  begin

```

1.5.1 More about the relation eq_app_right

```

lemma left_eq_app_right:
  "[u ∈ left_lang q; v ∈ left_lang q] ⇒ (u,v) ∈ eq_app_right language"
  ⟨proof⟩

lemma eq_app_right_class_eq:
  "UNIV // eq_app_right language = (λq. eq_app_right language ‘ ‘ {path_to
q}) ‘ accessible"
  ⟨proof⟩

lemma inj_right_lang_imp_eq_app_right_index:
  assumes "inj_on right_lang (dfa.states M)"
  shows "bij_betw (λq. eq_app_right language ‘ ‘ {path_to q})
          accessible (UNIV // eq_app_right language)"
  ⟨proof⟩

definition min_states where
  "min_states ≡ card (UNIV // eq_app_right language)"

lemma minimal_imp_index_eq_app_right:

```

```
"minimal  $\implies$  card (dfa.states M) = min_states"
<proof>
```

A minimal machine has a minimal number of states, compared with any other machine for the same language.

```
theorem minimal_imp_card_states_le:
  "[[minimal; dfa M'; dfa.language M' = language]]
 $\implies$  card (dfa.states M)  $\leq$  card (dfa.states M')]"
<proof>
```

```
definition index_f :: "'a list set  $\Rightarrow$  hf" where
  "index_f  $\equiv$  SOME h. bij_betw h (UNIV // eq_app_right language) (hfset
(ord_of min_states))"
```

```
lemma index_f: "bij_betw index_f (UNIV // eq_app_right language) (hfset
(ord_of min_states))"
<proof>
```

```
interpretation Canon:
  MyhillNerode_dfa language "eq_app_right language"
  language
  min_states index_f
<proof>
```

```
interpretation MN: dfa Canon.DFA
<proof>
```

```
definition iso :: "hf  $\Rightarrow$  hf" where
  "iso  $\equiv$  index_f o ( $\lambda$ q. eq_app_right language ‘ ‘ {path_to q})"
```

```
theorem minimal_imp_isomorphic_to_canonical:
  assumes minimal
  shows "dfa_isomorphism M Canon.DFA iso"
<proof>
```

```
lemma states_PR [simp]:
  "dfa.states (nfa_hf.Power_dfa_hf (Reverse_nfa M)) = HF ‘ Pow (dfa.states
M)"
<proof>
```

```
lemma inj_on_right_lang_PR:
  assumes "dfa.states M = accessible"
  shows "inj_on (dfa.right_lang (nfa_hf.Power_dfa_hf (Reverse_nfa
M)))
  (dfa.states (nfa_hf.Power_dfa_hf (Reverse_nfa M)))"
<proof>
```

```
abbreviation APR :: "'x dfa_hf  $\Rightarrow$  'x dfa_hf" where
  "APR X  $\equiv$  dfa.Accessible_dfa (nfa_hf.Power_dfa_hf (Reverse_nfa X))"
```

```

theorem minimal_APR:
  assumes "dfa.states M = accessible"
  shows "dfa.minimal (APR M)"
  <proof>

definition Brzozowski :: "'a dfa_hf" where
  "Brzozowski  $\equiv$  APR (APR M)"

lemma dfa_Brzozowski: "dfa_hf Brzozowski"
  <proof>

theorem language_Brzozowski: "dfa.language Brzozowski = language"
  <proof>

theorem minimal_Brzozowski: "dfa.minimal Brzozowski"
  <proof>

end

lemma index_f_cong:
  "[[dfa.language M = dfa.language N; dfa M; dfa N]]  $\implies$  dfa_hf.index_f
M = dfa_hf.index_f N"
  <proof>

theorem minimal_imp_isomorphic:
  "[[dfa.language M = dfa.language N; dfa.minimal M; dfa.minimal N;
dfa_hf M; dfa_hf N]]
 $\implies$   $\exists$ h. dfa_isomorphism M N h"
  <proof>

end

```